EXAMPLE 2.9-1: Fiber Optic Bundle

Lighting represents one of the largest uses of electrical energy in residential and commercial buildings; lighting loads are highest during on-peak hours when electrical energy is most costly. Also, the thermal energy deposited into conditioned space by electrical lighting adds to the air conditioning load on the building which, in turn, adds to the electrical energy required to run the air conditioning system.

A novel lighting system consists of a sunlight collector and a light distribution system, as shown in Figure 1. The sunlight collector tracks the sun and collects and concentrates solar radiation. The light distribution system receives the concentrated solar radiation and distributes it into a building where it is finally dispensed in fixtures that are referred to as luminaires. Sunlight contains both visible and invisible energy; only the visible portion of the sunlight is useful for lighting and therefore the collector gathers the visible portion of the incident solar radiation while eliminating the invisible ultraviolet and infrared portions of the spectrum. (We will learn more about these characteristics of radiation in Chapter 10).



Figure 1: Hybrid lighting system (Cheadle, 2006).

The fiber optic bundle used to transmit the visible light is composed of many, small diameter optical fibers that are packed in approximately a hexagonal close-packed array. A conductive filler material is wrapped around each fiber and the entire structure is simultaneously heated and compressed so that the fibers becomes hexagonal shaped with a thin layer of conductive filler separating each hexagon (Figure 2). The dimension of each face of the hexagon is d = 1.0 mm and the thickness of the filler that separates the hexagons is a = 50 μm thick. The fiber conductivity is k_{fb} = 1.5 W/m-K while the filler conductivity is k_{fl} = 50.0 W/m-K.



Figure 2: Array of optical fibers packed together and compressed in order to form a fiber optic bundle that has hexagonal units.

a.) Determine the effective radial and axial conductivity associated with the bundle.

The inputs to the problem related to the composite structure are entered in EES:

```
"EXAMPLE 2.9-1: Fiber Optic Bundle"
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Composite Structure Inputs"
d=1 [mm]*convert(mm,m)                 "face dimension of hexagon"
a=0.05 [mm]*convert(mm,m)              "filler thickness"
k_fl=50 [W/m-K]                        "conductivity of filler"
k_fb=1.5 [W/m-K]                       "conductivity of fiber"
```

The heat transfer in the axial direction can travel through two parallel paths, the filler and the fiber. The area of a single fiber (a hexagon with each side having length *d*) is:

$$A_{fb} = 2\, d^2 \sin\left(\frac{\pi}{3}\right)\left[1+\cos\left(\frac{\pi}{3}\right)\right]$$

```
A_fb=2*d^2*sin(pi/3)*(1+cos(pi/3))     "area of a fiber"
```

The heat transfer through a unit length of the fiber in the axial direction for a given temperature difference (ΔT) is:

$$\dot{q}_{fb} = \frac{\Delta T\, k_{fb}\, A_{fb}}{L}$$

The area of the filler material associated with a single fiber (which has thickness *a*/2 due to sharing of the filler with the neighboring fibers) is:

$$A_{fl} = 3\,d\,a$$

```
A_fl=3*d*a                    "area of filler associated with a fiber"
```

The heat transfer through a unit length of the filler in the axial direction for the same temperature difference is:

$$\dot{q}_{fl} = \frac{\Delta T\, k_{fl}\, A_{fl}}{L}$$

The equivalent homogenous material will have conductivity $k_{eff,x}$ and area $A_{fl} + A_{fb}$; therefore, the heat transfer through the equivalent material ($\dot{q}_{eff,x}$) is:

$$\dot{q}_{eff,x} = \frac{\Delta T\, k_{eff,x} \left(A_{fl} + A_{fb}\right)}{L}$$

The effective conductivity is defined so that the heat transfer through the equivalent material is equal to the sum of the heat transfer through the fiber and the filler:

$$\underbrace{\frac{\Delta T\, k_{eff,x}\left(A_{fl} + A_{fb}\right)}{L}}_{\dot{q}_{eff,x}} = \underbrace{\frac{\Delta T\, k_{fb}\, A_{fb}}{L}}_{\dot{q}_{fb}} + \underbrace{\frac{\Delta T\, k_{fl}\, A_{fl}}{L}}_{\dot{q}_{fl}}$$

or, solving for $k_{eff,x}$:

$$k_{eff,x} = \frac{k_{fb}\, A_{fb} + k_{fl}\, A_{fl}}{A_{fl} + A_{fb}}$$

The effective conductivity in the axial direction is the area-weighted conductivity of the two parallel paths:

```
k_eff_x=(k_fb*A_fb+k_fl*A_fl)/(A_fl+A_fb)          "effective conductivity"
```

which leads to $k_{eff,x}$ = 4.1 W/m-K.

The radial conductivity cannot be evaluated using a simple parallel or series resistance circuit because the heat flow across the bundle is complex and 2-D. Therefore, a 2-D finite element model of a unit cell of the structure (shown in Figure 2) must be generated. Figure 3 illustrates the details of a unit cell and includes the coordinates of the points (in mm) that define the geometry.



Figure 3: A unit cell of the fiber optic bundle structure; the points that define the structure are shown (dimensions are in mm).

The finite element model is generated using FEHT. A grid is specified where 1 cm of screen dimension corresponds to 0.2 mm. The five regions in Figure 3 are generated using five outlines; the points are initially placed approximately and then precisely positioned by double-clicking on each one in turn. The conductivity for regions 1 through 4 are set to 1.5 W/m-K (consistent with the optical fiber) and the conductivity for region 5 is set to 50.0 W/m-K (consistent with the filler material). The boundary conditions along the upper and lower edges are set as adiabatic. In order to set a temperature difference across the unit cell (from left to right) it would seem logical to set the temperature at the left hand side to 1.0°C and the right hand side to 0.0°C. However, due to the manner in which finite element techniques determines the heat flux at a surface, a more accurate answer is obtained if a convective boundary condition is set with a very high heat transfer coefficient; for example, 1×10^5 W/m^2-K.

A relatively crude mesh is generated and then refined, particularly in the filler material where most of the heat flow is expected. The result is shown in Figure 4(a). The finite element model is solved and the temperature contours are shown in Figure 4(b).





(a) **(b)**

Figure 4: Finite element model showing (a) the refined grid and (b) the temperature distribution.

The heat flow through the unit cell can be determined by selecting Heat Flows from the View menu and then selecting all of the boundaries on either the left or right hand side. The selection process can be accomplished by using the mouse to 'drag' a selection rectangle around the lines on the boundary. The total heat flow is $\dot{q}_{eff,r}/L$ = 3.070 W/m; this result can be used to compute the effective conductivity of the composite. The effective conductivity across the bundle ($k_{eff,r}$) is defined as the conductivity of a homogeneous material that would provide the same heat transfer per length into the page as the composite structure simulated by the finite element model.

$$\frac{\dot{q}_{eff,r}}{L} = k_{eff,r}\frac{H\,\Delta T}{W}$$

where H is the height of the unit cell (1.782 mm, from Figure 3), W is the width of the unit cell (1.558 mm, from Figure 3), and ΔT = 1.0 K (the imposed boundary condition).

```
"k_eff_r calculation using inputs from FEHT"
H=1.782 [mm]*convert(mm,m)          "height of unit cell"
W=1.558 [mm]*convert(mm,m)          "width of unit cell"
q_dot_eff_r\L=3.070 [W/m]           "rate of heat transfer per unit length predicted by FE model"
DT=1 [K]                            "temperature difference imposed on FE model"
q_dot_eff_r\L=k_eff_r*H*DT/W        "effective conductivity in the radial direction"
```

which leads to $k_{eff,r}$ = 2.7 W/m-K.

The effective conductivity in the x-direction is 4.1 W/m-K while the effective conductivity in the radial direction is 2.7 W/m-K. These values make sense; both lie between the conductivity of the fiber and the filler and both are closer to the conductivity of the fiber because it occupies most of the space. Further, the conductivity in the x-direction is larger because the path through the high conductivity filler is more direct in this direction.

The advantage of the hexagonal pattern in Figure 2 is that is has the lowest possible porosity (ϕ). The porosity is defined as the fraction of the area of the face of the bundle that is occupied by the filler. The optical fibers are designed so that any radiation that strikes the face of a fiber is "trapped" by total internal reflection and transmitted without substantial loss. However, the radiation that strikes the opaque filler in the interstitial areas between the fibers will be absorbed and result in a thermal load on the bundle that manifests itself as a heat flux on the surface. This heat flux can lead to elevated temperatures and thermal failure.

Assume that the outer edge of the fiber optic bundle is exposed to air at T_∞ = 20ºC with a heat transfer coefficient $\overline{h}_{out}$ = 5.0 W/m^2-K. The front face of the bundle (at x = 0) is exposed to air at T_∞ = 20ºC with a heat transfer coefficient $\overline{h}_f$ = 10.0 W/m^2-K. The radius of the bundle is r_{out} = 2.0 cm and its length is L = 2.5 m. The end of the bundle at $x = L$ is maintained at a temperature of T_L = 20ºC. A schematic of the problem is shown in Figure 5.



Figure 5: Schematic of the fiber optic bundle problem.

b.) Is it appropriate to treat the bundle as an extended surface? Justify your answer.

The inputs for the problem are entered in EES:

```
"Problem Inputs"
r_out=2 [cm]*convert(cm,m)                "outer radius"
L = 2.5 [m]                               "bundle length"
h_bar_out=5 [W/m^2-K]                     "heat transfer coefficient on outer surface"
T_infinity=converttemp(C,K,20[C])         "air temperature"
h_bar_f=10 [W/m^2-K]                      "heat transfer coefficient on the face"
T_L=converttemp(C,K,20[C])                "temperature at x=L"
q_dot_flux_inc=100000 [W/m^2]             "incident radiant heat flux on the face"
```

The extended surface approximation neglects temperature gradients in the radial direction within the bundle. The conduction resistance in the radial direction ($R_{cond,r}$) must be neglected in order to treat the bundle as an extended surface. This assumption is justified provided that $R_{cond,r}$ is small relative to the resistance that is being considered, convection from the outer surface of the bundle (R_{conv}). The appropriate Biot number is therefore:

$$Bi = \frac{R_{cond,r}}{R_{conv}}$$

It is not possible to precisely compute a resistance that characterizes conduction in the radial direction within the bundle; conduction from the center of the bundle would be characterized by an infinite resistance. Instead, an approximate conduction length (r_{out} /2) and area ($\pi\, r_{out}\, L$) are used:

$$Bi = \underbrace{\left(\frac{r_{out}}{2\, k_{eff,r}\, \pi\, r_{out}\, L}\right)}_{R_{cond,r}} \underbrace{\left(\frac{\bar{h}_{out}\, 2\, \pi\, r_{out}\, L}{1}\right)}_{R_{conv}} = \frac{\bar{h}_{out}\, r_{out}}{k_{eff,r}}$$

```
Bi=h_bar_out*r_out/k_eff_r                "Biot number"
```

The Biot number is 0.037 which is much less than 1 and therefore the extended surface approximation is justified.

c.) Develop an analytical model for the temperature distribution in the bundle.

The differential control volume used to derive the governing equation is shown in Figure 5 and leads to the energy balance:

$$\dot{q}_x = \dot{q}_{x+dx} + \dot{q}_{conv}$$

which can be expanded and simplified:

$$0 = \frac{d\dot{q}_x}{dx} dx + \dot{q}_{conv} \tag{1}$$

The rate equations are:

$$\dot{q}_x = -k_{eff,x} \, \pi \, r_{out}^2 \frac{dT}{dx} \tag{2}$$

$$\dot{q}_{conv} = 2 \, \pi \, r_{out} \, \overline{h}_{out} \, dx \left(T - T_\infty\right) \tag{3}$$

Substituting Eqs. (1) and (2) into Eq. (3) leads to:

$$0 = \frac{d}{dx}\left[-k_{eff,x} \, \pi \, r_{out}^2 \frac{dT}{dx}\right] dx + 2 \, \pi \, r_{out} \, \overline{h}_{out} \, dx \left(T - T_\infty\right)$$

or

$$\frac{d^2 T}{dx^2} - m^2 \, T = -m^2 \, T_\infty \tag{4}$$

where

$$m^2 = \frac{2 \, \overline{h}_{out}}{k_{eff,x} \, r_{out}}$$

The governing differential equation, Eq. (4), is satisfied by exponential functions. The differential equation can also be entered in Maple and solved.

```
> restart;
> ODE:=diff(diff(T(x),x),x)-m^2*T(x)=-m^2*T_infinity;
```

$$ODE := \left(\frac{d^2}{dx^2}T(x)\right) - m^2\,T(x) = -m^2\,T\_infinity$$

```
> Ts:=dsolve(ODE);
```

$$Ts := T(x) = e^{(-m\,x)}\,\_C2 + e^{(m\,x)}\,\_C1 + T\_infinity$$

The solution is copied and pasted into EES.

```
"Solution"
m=sqrt(2*h_bar_out/(k_eff_x*r_out))                       "solution parameter"
T = exp(-m*x)*C_2+exp(m*x)*C_1+T_infinity                 "solution from Maple"
```

The first boundary condition is the specified temperature at x=L:

$$T_{x=L} = T_L$$

A symbolic expression for this boundary condition is obtained in Maple:

```
> rhs(eval(Ts,x=L))=T_L;
```

$$e^{(-m\,L)}\,\_C2 + e^{(m\,L)}\,\_C1 + T\_infinity = T\_L$$

and pasted into EES:

```
exp(-m*L)*C_2+exp(m*L)*C_1+T_infinity = T_L  "boundary condition at x=L"
```

The second boundary condition is obtained from an interface energy balance at x=0. Recall that only the flux incident on the filler material results in a heat load; therefore, the incident heat flux is multiplied by the porosity (ϕ), which is the ratio of the area of the filler to the total area:

$$\phi = \frac{A_{fl}}{A_{fl} + A_{fb}}$$

```
phi=A_fl/(A_fl+A_fb)                                      "porosity"
```

The absorbed flux must either be transferred by conduction to the bundle or convection to the air at the face:

$$\dot{q}''_{inc}\,\phi = -k_{eff,x}\left.\frac{dT}{dx}\right|_{x=0} + \bar{h}_f\left(T_{x=0} - T_{air}\right)$$

A symbolic expression for this boundary condition is obtained in Maple:

```
> q_dot_flux_inc*phi=-k_eff_x*rhs(eval(diff(Ts,x),x=0))+h_bar_f*(rhs(eval(Ts,x=0))-T_infinity);
```

$$q\_dot\_flux\_inc\ \phi = k\_eff\_x\,(-m\,\_C2 + m\,\_C1) + h\_bar\_f\,(\_C2 + \_C1)$$

and pasted into EES:

```
q_dot_flux_inc*phi = -k_eff_x*(-m*C_2 + m*C_1) + h_bar_f*(C_2 + C_1)
        "boundary condition at x=0"
```

The solution is converted to Celsius:

```
T_C=converttemp(K,C,T)        "temperature in C"
```

Figure 6 illustrates the temperature distribution near the face of the fiber optic bundle.



Figure 6: Temperature distribution in fiber optic bundle

The model can be used to assess alternative methods of thermal management. For example, the heat transfer coefficient at the face might be increased with fan cooling or the conductivity of the filler material might be increased through material selection. Figure 7 illustrates the maximum temperature (the temperature at the face) as a function of $\bar{h}_f$ for various values of k_f.



Figure 7: Temperature at the face as a function of the face heat transfer coefficient for various values of the filler material conductivity.